

FAA TECHNICAL CENTER LETTER REPORT

RECEIVED

MAR 22 1982

TECHNICAL CENTER
ATLANTIC CITY, N.J. 08403

DIGITIZING PLOTTING PROCEDURE FOR
MICROWAVE LANDING SYSTEM FLIGHT TEST
AIRBORNE AND GROUND TRACKER MERGED
DATA PLOTS

by

Bruce Hendrick

MARCH 1982

U. S. DEPARTMENT OF TRANSPORTATION
FEDERAL AVIATION ADMINISTRATION
TECHNICAL CENTER
Atlantic City Airport, N.J. 08403

TABLE OF CONTENTS

ABSTRACT

PURPOSE

BACKGROUND

DISCUSSION

APPENDIX

- Program Flowchart
- Program Listing
- Questions and Answers with Sample Plots

ABSTRACT

This program was written to apply tolerance limits to Microwave Landing System (MLS) flight test airborne and ground tracker merged data plots. It will be used by project engineers for post flight analysis and determination of acceptable data. A program flow chart, a program listing, and sample plots are presented in the Appendix.

Project No. 076-320-110

Key Words

MLS
GDU'S
UDU'S
Calcomp plot

PURPOSE

The purpose of the plotting technique revision was to significantly increase the efficiency and ease by which tolerances are plotted onto original Calcomp plots for Microwave Landing System (MLS) test flight data by project personnel.

BACKGROUND

Previously, tolerances were plotted with the Hewlett-Packard System 9854B and accompanying H.P. 9872A plotter onto Calcomp plots only after the following 3 steps were taken:

Step 1. Locate and plot on a clean sheet of paper, the axes by which the computer will plot the tolerances.

Step 2. Carefully align the plot to be processed so that it's axes coincide with those of Step 1.

Step 3. Run the program to plot the tolerances.


Typically, Step 2 took approximately 1 to 1½ minutes to perfectly align the two sets of axes. For example, to process any given flight data that had roughly 10 runs (40 plots) associated with it, the time it took to plot the tolerances was around one hour. This time factor, together with the fact that the condition of the paper with the axes generated in Step 1, gradually diminished as the plotting process continued, led to the search for a better, more accurate, and more efficient plotting technique.


DISCUSSION


The main benefit of this technique revision is to orient the Calcomp plot in any position on the H.P. plotter, thus eliminating the need to align the plot relative to the plotting surface, and still retain sufficient accuracy. Consequently, there was a need to devise a way to rotate and translate the plot relative to a known point (0,0) on the plotter. This was best solved by performing all plotting in Graphic Defined Units (GDU's) as opposed to User Defined Units (UDU's). In this way, the coordinates of the points to be plotted can be found with respect to a known, unchangeable position.


By digitizing the top of the Y axis, the bottom of the Y axis, and the end of the X axis in GDU's, a correction factor in GDU's per UDU for each axis is calculated. The angle of rotation and the distance in X and Y from the plotter's (0,0) point (lower left-hand corner) are also found. The coordinates of the Calcomp plots' rotated and translated origin are then computed. At this point, the previously calculated coordinates of the tolerances in UDU's are converted to GDU's, rotated by the angle of rotation, and translated to their respective positions relative to the new origin. The program then plots the revised tolerances and asks the user to prepare the next plot.


This procedure is just as accurate as the previous one while reducing the time needed to process plots by eliminating the need to align the papers. The time required to process any given series of runs is reduced in half using the new technique.





























```

10      ! ***** "ROTDCA" *****
20      ! *****
30      ! ^^ PLOTTING TOLERANCES ONTO DISPLACED AND ROTATED AXES ^^
40      ! *** PROGRAM "ROTDCA" PLOTS MLS PFE AND CMN TOLERANCES FOR AZ AND EL
50      ! *** FOR WASHINGTON NATIONAL AIRPORT. ***
60      ! *** WRITTEN BY B. HENDRICK ON 1/28/82 ***
70      ! *****
80      !
90      ! FILE "WASH" HOLDS DATA FOR DISTANCES FROM AZ PHASE CENTER FOR RADIALS
100     ! AND FILE "DSTORE" HOLDS THE ROUTINE TO ALTER THIS DATA
110     !
120     ! FILE "DIST2" HOLDS DATA FOR DISTANCES FROM AZ PHASE CENTER FOR APPROACHE
130     ! AND FILE "DSTOR" HOLDS THE ROUTINE TO ALTER THIS DATA
140     !
150     !
160     OPTION BASE 1
170     DIM Psi(3)
180     DIM D(25)
190     PLOTTER IS 7,5,"9872A"
200     PRINTER IS 16           !          ENABLES THE CRT
210     PRINT "ENTER TYPE OF FLIGHT: RADIAL=1, APPROACH=2, ORBIT=3"
220     INPUT L
230     PRINT L,LIN(1)
240     IF L=2 THEN 290
250     IF L=3 THEN 320
260     ASSIGN #1 TO "WASH"
270     S=25
280     GOTO 360
290     ASSIGN #1 TO "DIST2"
300     S=10
310     GOTO 360
320     Psi(1)=-10
330     Psi(2)=0 ! SET AZ TRACKER VALUES FOR ORBIT ONLY
340     Psi(3)=10
350     GOTO 390
360     FOR B=1 TO S
370     READ #1;D(B)
380     NEXT B
390     IF L=2 THEN 440
400     PRINT "ENTER ALTITUDE (IN FEET)"
410     INPUT Alt1
420     PRINT Alt1,LIN(1)
430     Alt=Alt1/6076           ! CONVERTS FEET TO NM
440     PRINT "ENTER 1 IF YOU DON'T WANT ALL FOUR TYPES OF PLOTS"
450     INPUT No
460     PRINT No,LIN(1)
470     IF L=1 THEN 520           ! RADIAL
480     IF L=3 THEN 560           ! ORBIT
490     PRINT "ENTER APPROACH ELEVATION ANGLE (GLIDE PATH)"
500     INPUT Phi
510     PRINT Phi,LIN(1)
520     PRINT "ENTER AZ ANGLE"
530     INPUT Psi
540     PRINT Psi,LIN(1)
550     GOTO 610
560     PRINT "ENTER ORBIT DIST FROM AZ PHASE CENTER(NM)"
570     INPUT L1
580     PRINT L1,LIN(1)
590     IF L=3 THEN H=3
600     GOTO 630
610     IF L=1 THEN H=25
620     IF L=2 THEN H=10
630     IF No=0 THEN 680
640     PRINT "ENTER TYPE OF PLOT:1=AZ PFE, 2=AZ CMN, 3=EL PFE, 4=EL CMN,(ENTER 1
2 FOR BOTH AZ PFE AND CMN, 34 FOR BOTH EL PFE AND CMN)"

```

```

650 INPUT K3
660 PRINT K3,LIN(1)
670 GOTO 690
680 K3=1
690 PRINT "PAPER COPY? 1=YES"
700 INPUT K2
710 PRINT K2,LIN(1)
720 IF K2<>1 THEN 800
730 PRINTER IS 0
740 PRINT "AZ PFE","AZ CMN","EL PFE","EL CMN",LIN(3)
750 !
760 !
770 ! TOLERANCE CALCULATIONS:
780 !
790 !
800 Add=0
810 X=4954/6076 !DIST FROM AZ ANTENNA TO A POINT ON RWY OPPOSITE EL ANTENNA
820 X1=5212/6076 ! DIST FROM AZ ANTENNA TO THRESHOLD
830 X2=.22
840 X3=.133
850 X4=.05
860 X5=.1
870 X6=.5
880 DIM A(25,4)
890 DEG
900 !
910 ! TRANSFER OF REFERENCE POINT FROM AZ ANTENNA TO RWY
920 !
930 FOR B=1 TO H
940 IF L=3 THEN 980 ! SKIP IF ORBIT
950 Y=D(B)*SIN(Psi)
960 X7=D(B)*COS(Psi)-X
970 GOTO 1000
980 Y=L1*SIN(Psi(B))
990 X7=L1*COS(Psi(B))
1000 Z=Y/X7
1010 Theta=ATN(Z) !THETA= AZ ANGLE FROM RWY REFERENCE DATUM
1020 !
1030 ! ***** AZ TOLERANCES *****
1040 !
1050 IF L=2 THEN 1130
1060 IF L=1 THEN 1090
1070 Z1=A1t/L1
1080 GOTO 1100
1090 Z1=A1t/D(B)
1100 Phi1=ATN(Z1) !Phi1=EL ANGLE FROM AZ ANTENNA
1110 IF Phi1<9 THEN 1130
1120 GOTO 1140
1130 Phi1=9
1140 IF Phi1>20 THEN 1670
1150 !
1160 ! PFE CALCULATIONS
1170 !
1180 IF L=3 THEN 1210
1190 Dfeaz=1+(D(B)-X1)/20
1200 GOTO 1220
1210 Dfeaz=1+(L1-X1)/20
1220 Afeaz=1+ABS(.5*Theta/40)
1230 Efeaz=1+(Phi1-9)/11
1240 Pfeaz=X2*Dfeaz*Afeaz*Efeaz ! ***** AZ PFE *****
1250 !
1260 ! CMN CALCULATIONS
1270 !
1280 IF L=3 THEN 1310
1290 Dmnaz=1+.3*(D(B)-X1)/10
1300 GOTO 1320

```

```

1310 Dmnaz=1+.3*(L1-X1)/10
1320 Amnaz=1+ABS(.3*Theta/40)
1330 Cmnaz=X5*Dmnaz*Amnaz ! **** AZ CMN ****
1340 !
1350 !      $$$$$$ EL TOLERANCES $$$$$$
1360 !
1370 IF L=2 THEN 1440
1380 IF L=1 THEN 1410
1390 Z1=Alt/(L1-X)
1400 GOTO 1430
1410 E=D(B)-X
1420 Z1=Alt/E
1430 Phi=ATN(Z1)
1440 IF Phi<6 THEN 1460 ! Phi=GLIDE PATH ANGLE
1450 GOTO 1470
1460 Phi=6
1470 !
1480 !      PFE CALCULATIONS
1490 !
1500 IF L=3 THEN 1530
1510 Dfeel=1+X6*(D(B)-X1)/20
1520 GOTO 1540
1530 Dfeel=1+X6*(L1-X1)/20
1540 Efeel=1+(Phi-6)/9
1550 Pfeel=X3*Dfeel*Efeel*Amnaz ! **** EL PFE ****
1560 !
1570 !      CMN CALCULATIONS
1580 !
1590 Cmnel=X4*Dmnaz*Amnaz*Efeel ! **** EL CMN ****
1600 !
1610 A(B,1)=Pfeaz
1620 A(B,2)=Cmnaz
1630 A(B,3)=Pfeel
1640 A(B,4)=Cmnel
1650 !
1660 PRINT A(B,1),A(B,2),A(B,3),A(B,4)
1670 NEXT B
1680 IF L=3 THEN 1720
1690 PRINT "DIST VALUES(INPUT)"
1700 MAT PRINT D
1710 GOTO 1750
1720 PRINT "AZ ANGLE VALUES(INPUT)"
1730 MAT PRINT Psi
1740 !
1750 !      ***** SET UP PLOTTER *****
1760 !
1770 SETGU
1780 PRINT "DIGITIZE TOP OF Y AXIS"
1790 POINTER 11,100
1800 DIGITIZE X1,Y1
1810 PRINT
1820 PRINT "DIGITIZE BOTTOM OF Y AXIS"
1830 POINTER 11,5
1840 DIGITIZE X2,Y2
1850 PRINT
1860 PRINT "DIGITIZE END OF X AXIS"
1870 POINTER 115,5
1880 DIGITIZE X3,Y3
1890 Xdis=X3-X2
1900 Ydis=Y1-Y2
1910 IF L=3 THEN Xgupuu=Xdis/20
1920 IF L<>3 THEN Xgupuu=Xdis/10
1930 Ygupuu=Ydis/.90
1940 DEG
1950 Rot=ATN((X1-X2)/(Y1-Y2))
1960 IF L=3 THEN X0=(X1+X3)/2

```

```

1970 IF L=3 THEN Y0=(Y1+Y3)/2
1980 IF L<>3 THEN X0=(X1+X2)/2
1990 IF L<>3 THEN Y0=(Y1+Y2)/2
2000 LOCATE 0,120,0,100
2010 !
2020 !
2030 ! *****PLOTTING OF TOLERANCES *****
2040 !
2050 !
2060 FOR W=2 TO 3
2070 FOR B=1 TO H
2080 IF L<>3 THEN Xer=D(B)*Xgupuu
2090 IF L=3 THEN Xer=Psi(B)*Xgupuu
2100 Y1er=A(B,1)*Ygupuu*(-1)^W
2110 Y2er=A(B,2)*Ygupuu*(-1)^W
2120 Y3er=A(B,3)*Ygupuu*(-1)^W
2130 Y4er=A(B,4)*Ygupuu*(-1)^W
2140 X1r=Xer*COS(Rot)+Y1er*SIN(Rot)
2150 X2r=Xer*COS(Rot)+Y2er*SIN(Rot)
2160 X3r=Xer*COS(Rot)+Y3er*SIN(Rot)
2170 X4r=Xer*COS(Rot)+Y4er*SIN(Rot)
2180 Y1r=Y1er*COS(Rot)-Xer*SIN(Rot)
2190 Y2r=Y2er*COS(Rot)-Xer*SIN(Rot)
2200 Y3r=Y3er*COS(Rot)-Xer*SIN(Rot)
2210 Y4r=Y4er*COS(Rot)-Xer*SIN(Rot)
2220 X1new=X1r+X0
2230 X2new=X2r+X0
2240 X3new=X3r+X0
2250 X4new=X4r+X0
2260 Y1new=Y1r+Y0
2270 Y2new=Y2r+Y0
2280 Y3new=Y3r+Y0
2290 Y4new=Y4r+Y0
2300 IF K3=34 THEN 2430
2310 IF K3=12 THEN 2340
2320 IF K3>1 THEN 2370
2330 LINE TYPE 1
2340 IF (A(B,1)<=.45) AND (A(B,1)<>0) THEN PLOT X1new,Y1new! PLOTS AZ PFE
2350 GOTO 2480
2360 IF K3=12 THEN 2380
2370 IF K3>2 THEN 2410
2380 LINE TYPE 5
2390 IF (A(B,2)<=.45) AND (A(B,2)<>0) THEN PLOT X2new,Y2new! PLOTS AZ CMN
2400 GOTO 2480
2410 IF K3=34 THEN 2430
2420 IF K3>3 THEN 2460
2430 LINE TYPE 1
2440 IF (A(B,3)<=.45) AND (A(B,3)<>0) THEN PLOT X3new,Y3new! PLOTS EL PFE
2450 GOTO 2480
2460 LINE TYPE 5
2470 IF (A(B,4)<=.45) AND (A(B,4)<>0) THEN PLOT X4new,Y4new! PLOTS EL CMN
2480 NEXT B
2490 PENUP
2500 NEXT W
2510 PENUP
2520 IF K3=12 THEN 2630
2530 IF K3=34 THEN 2760
2540 !
2550 ! ***** CONTINUE PLOTTING WITH SAME DATA *****
2560 ! +++ ADVANCE TO NEXT PLOT +++
2570 !
2580 IF No=1 THEN 2820
2590 IF (No=0) AND (K3=1) THEN 2630
2600 IF (No=0) AND (K3=2) THEN 2700
2610 IF (No=0) AND (K3=3) THEN 2760
2620 IF (No=0) AND (K3=4) THEN 2820

```

```
2630  PRINTER IS 16
2640  PRINT "IS PAPER READY FOR AZ CMN PLOT<FILT 2>? 2=STOP PROGRAM"
2650  BEEP
2660  INPUT Cont1
2670  IF Cont1=2 THEN 2820
2680  K3=2
2690  GOTO 1750
2700  PRINT "IS PAPER READY FOR EL PFE PLOT<FILT 1>? 2=STOP PROGRAM"
2710  BEEP
2720  INPUT Cont2
2730  IF Cont2=2 THEN 2820
2740  K3=3
2750  GOTO 1750
2760  PRINT "IS PAPER READY FOR EL CMN PLOT<FILT 2>? 2=STOP PROGRAM"
2770  BEEP
2780  INPUT Cont3
2790  IF Cont3=2 THEN 2820
2800  K3=4
2810  GOTO 1750
2820  PRINT "END OF PROGRAM"
2830  BEEP
2840  STOP
2850  END
```

** QUESTIONS AND ANSWERS FOR A 6 DEGREE APPROACH **

PLOTS ONE THROUGH FOUR ARE THE RESULTS OF PLOTTING THE DATA BELOW

ENTER TYPE OF FLIGHT: RADIAL=1, APPROACH=2, ORBIT=3

2

ENTER 1 IF YOU DON'T WANT ALL FOUR TYPES OF PLOTS

0

ENTER APPROACH ELEVATION ANGLE (GLIDE PATH)

6

ENTER AZ ANGLE

0

PAPER COPY? 1=YES

1

AZ PFE	AZ CMN	EL PFE	EL CMN
.222664186965	.100726596445	.133805311061	5.03632982225E-02
.232564186965	.103426596445	.136797811061	5.17132982225E-02
.243564186965	.106426596445	.140122811061	5.32132982225E-02
.254564186965	.109426596445	.143447811061	5.47132982225E-02
.265564186965	.112426596445	.146772811061	5.62132982225E-02
.276564186965	.115426596445	.150097811061	5.77132982225E-02
.287564186965	.118426596445	.153422811061	5.92132982225E-02
.298564186965	.121426596445	.156747811061	6.07132982225E-02
.309564186965	.124426596445	.160072811061	6.22132982225E-02
.320564186965	.127426596445	.163397811061	6.37132982225E-02

DIST VALUES(INPUT)

1.1	2	3	4
5	6	7	8
9	10	0	0
0	0	0	0
0	0	0	0
0	0	0	0
0			

DIGITIZE TOP OF Y AXIS


DIGITIZE BOTTOM OF Y AXIS

DIGITIZE END OF X AXIS

JUN 05 10 59 AMCH C RZ S BEG APP RUN 10
331 HPG NADY GMI
PFE 018 1

DATA PROCESSED BY ECG TECHNICAL CENTER
ATLANTIC CITY AIRPORT, NJ 08403

14 JUN 1982




NM FROM RZ PHASE CENTER

(1)

JAN 05 1982 08:00:00 AZ 5 0200 APP RUV 10
139E ARC NAD/GRM1
CMR SYS 1

DATA PROCESSED BY FAA TECHNICAL CENTER
ATLANTIC CITY AIRPORT, NJ 08405

14 JAN 1982




NM FROM 92 PHASE CENTER

(2)

JUN 05.1992 WASH. DC AZ 5 250 APP RUN 10
1352 HRS NWS/OMNI
PFE 075 I

DATA PROCESSED BY FSS TECHNICAL CENTER
ATLANTIC CITY AIRPORT, NJ 08405

34 JAN 1992




JAN 05 1982 WASH DC RZ 5 REC APP RUN 10
1352 HRS N42/0MMT
CMN SYS 1

DATA PROCESSED BY FSS TECHNICAL CENTER
ATLANTIC CITY AIRPORT, NJ 08405

14 JAN 1982

0.45
0.40
0.35
0.30
0.25
0.20
0.15
0.10
0.05
0.00
-0.05
-0.10
-0.15
-0.20
-0.25
-0.30
-0.35
-0.40

CHORLES
IN OR
DIF IN
DEL



F - FRAME FLAG
S - SYSTEM FLAG

0.00 1.00 2.00 3.00 4.00 5.00 6.00 7.00 8.00 9.00 10.00

NM FROM RZ FURSS CENTER

(4)

** QUESTIONS AND ANSWERS FOR A 4000' ORBIT **

PLOTS FIVE THROUGH EIGHT ARE THE RESULTS OF PLOTTING THE DATA BELOW

ENTER TYPE OF FLIGHT: RADIAL=1, APPROACH=2, ORBIT=3

3

ENTER ALTITUDE (IN FEET)

4000

ENTER 1 IF YOU DON'T WANT ALL FOUR TYPES OF PLOTS

0

ENTER ORBIT DIST FROM AZ PHASE CENTER(NM)

7

PAPER COPY? 1=YES

1

AZ PFE

AZ CMN

EL PFE

EL CMN

.323509710336

.127308591178

.166321983994

6.41917141988E-01

.287564186965

.118426596445

.154718124646

5.97182225105E-02

.323509710336

.127308591178

.166321983994

6.41917141988E-01

AZ ANGLE VALUES(INPUT)

-10

0

10

DIGITIZE TOP OF Y AXIS


DIGITIZE BOTTOM OF Y AXIS

DIGITIZE END OF X AXIS

JAN 05 1982 0858 7NM 014 084 4000' RUN 4
1236 HRS N23/ 08411
PFE 018
SYS 1

DATA PROCESSED BY FAA TECHNICAL CENTER
ATLANTIC CITY AIRPORT, NJ 08405

14 JAN 1982




AZ TRACKER IN DEGREES

(5)

JAN 05 1032 4534 7MM CCW DBA 4000' PUN 4
1230 HRS N45/ OMNI
OMN SYS 1

DATA PROCESSED BY EGG TECHNICAL CENTER
ATLANTIC CITY AIRPORT, NJ 08405


14 JAN 1982



JAN 05 1982 WASH TNM CCM ORA 4000 RUN 4
1236 HRS N427 GMMI
PFE 345 1

DATA PROCESSED BY ERA TECHNICAL CENTER
ATLANTIC CITY AIRPORT, NJ 08405

14 JAN 1982




AZ TRACKER IN DEGREES

JAN 05 1982 WASH 7MM CCR CBS 4000' RUN 4
1230 HRS NAO/OMNI
OMN 575 1

DATA PROCESSED BY FGR TECHNICAL CENTER
ATLANTIC CITY AIRPORT, NJ 08403

14 JAN 1982



RZ TRACKER IN DEGREES